
 

 

 

;clmlevprtp 
;avance-version 
;homonuclear Hartman-Hahn transfer using MLEV17 sequence 
;   for mixing 
;using two power levels for excitation and spinlock 
;phase sensitive using TPPI 
;A. Bax & D.G. Davis, J. Magn. Reson. 65, 355-360 (1985) 
;"clean-TOCSY" - sequence 
;C. Griesinger, G. Otting, K. Wuethrich & R.R. Ernst, 
;   J. Am. Chem. Soc. 110, 7870 (1988) 
; Parameter values…jSnyder…02-Mar-01 for DRX-400 
 
"p5=p6*.667" 
 
"p7=p6*2" 
 
"d0=3u" 
 
"d11=30m" 
 
"d12=20u" 
 
"d13=3u" 
 
 
1 ze  
2 d11  
  3m 
3 d12 pl9:f1 
  d1 cw:f1 ph29 
  d13 do:f1 
  d12 pl1:f1 
  p1 ph1 
  d0 
  d12 pl10:f1 
  (p17 ph26) 
4 (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph22 d20 p7 ph23 d20 p6 ph22) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p6 ph24 d20 p7 ph25 d20 p6 ph24) 
  (p5 ph23) 
  lo to 4 times l1  
  go=2 ph31 



 

 

  d11 wr #0 if #0 id0 ip1 zd 
  3m ip29 
  lo to 3 times td1 
exit  
   
ph1=0 2 2 0 1 3 3 1 
ph22=3 1 3 1 0 2 0 2 
ph23=0 2 0 2 1 3 1 3 
ph24=1 3 1 3 2 0 2 0 
ph25=2 0 2 0 3 1 3 1 
ph26=0 2 0 2 1 3 1 3 
ph29=0 
ph31=0 2 2 0 1 3 3 1 
 
 
;pl1 : f1 channel - power level for pulse (default)     [ -3 dB ] 
;pl9 : f1 channel - power level for presaturation      [ 55 dB ] 
;pl10: f1 channel - power level for TOCSY-spinlock     [ 9 dB  ] 100 to 120usec 360 pulse 
;p1 : f1 channel -  90 degree high power pulse 
;p5 : f1 channel -  60 degree low power pulse 
;p6 : f1 channel -  90 degree low power pulse        [25 to 30usec ] 
;p7 : f1 channel - 180 degree low power pulse 
;p17: f1 channel - trim pulse                         [2.5 msec] 
;d0 : incremented delay (2D)                         [3 usec] 
;d1 : relaxation delay; 1-5 * T1 
;d11: delay for disk I/O                             [30 msec] 
;d12: delay for power switching                      [20 usec] 
;d13: short delay                                    [3 usec] 
;d20: delay to allow compensation of NOE and ROE 
;     in theory d20 = p6, in practice d20 = 2.6 * p6 
;l1: loop for MLEV cycle:  
;       ((p6*64 + d20*32 + p5) * l1) + (p17*2) = mixing time 
;in0: 1/(2 * SW) = DW 
;nd0: 2 
;NS: 8 * n 
;DS: 16 
;td1: number of experiments 
;MC2: TPPI 
 
Look at filename:   A_pr-TOCSY in Franklin directory 
          expno 1 = 1D 
        expno 10 = 2D 


